

READING APPLEWORKS/3 EZ PIECES TEXTFILES FROM BASIC

By

David Ottalini
Washington Apple Pi
Apple /// SIG Co-Chairman

This is a tough one and I've decided to throw it out to you real "Business Basic Programmers" out there. I was able to find an AppleSoft program that will read AppleWorks or 3 EZP's Wordprocessing files. These are different from straight ASCII text files because AW/3EZPs adds information at the top of each file. There are plenty of Peeks and Pokes in this one, more than I can handle. So any help in converting this over to Business Basic would be appreciated (You GS Basic folks might want to give it a try as well). The program (with a tiny bit of conversion) looks like this:

```
10 REM Program by William A. Boyd
20 REM Donated to the Public Domain
30 REM Through the Atlanta Apple Users Group.
40 REM All Commercial Rights Reserved
50 REM Copyright (c) 1986
60 REM
70 REM This program demonstrates the logic of reading
80 REM Appleworks wordprocessing files from Basic.System
90 REM
100 ONERR GOTO 20000
200 TEXT : HOME : VPOS= 5
210 PRINT "This program will show you how to work"
220 PRINT "with AppleWorks Word Processing files"
230 PRINT "from Applesoft Basic Programs."
240 PRINT : PRINT "Select a small word processing file"
250 PRINT "and enter the filename here: ";
270 INPUT " ";FL$
300 REM BS= File starting point
310 BS = 8192:HL = 300
320 PRINT CHR$(4);"Blod";FL$;"A";BS;"TAWP"
330 HOME : SPEED= 70: INVERSE
340 PRINT "DISPLAY FILE: ";FL$: NORMAL : POKE 34,1
345 REM Lines 360-400 form MAIN BODY program
```

```

350 START = BS + HL
360 Y = PEEK (START + 1)
370 IF Y > = 208 THEN GOSUB 10000
380 IF Y < 208 THEN GOSUB 19000
400 GOTO 360
10000 REM Sub handles <CR> lines & Command Lines
10010 IF Y = 208 THEN PRINT CHR$ (13): GOTO 11000
10020 IF Y = 223 THEN PRINT "----justify----": GOTO 11000
10030 IF Y = 224 THEN PRINT "----unjustify---": GOTO 11000
10040 IF Y = 225 THEN PRINT "----center----": GOTO 11000
10050 IF Y = 255 THEN POP : SPEED= 255: TEXT : END
10060 IF Y = 230 THEN PRINT "----single space----": GOTO 11000
10070 IF Y = 231 THEN PRINT "----double space----": GOTO 11000
10080 IF Y = 232 THEN PRINT "----triple space----": GOTO 11000
10090 IF Y = 233 THEN PRINT "----new page ----": GOTO 11000
11000 START = START + 2: RETURN
19000 REM Subroutine handles Standard Text
19010 SL = PEEK (START): REM Length of string
19020 Z = ( PEEK (START + 3) > 127)
19030 FOR I = START + 4 TO START + SL + 1
19035 IF PEEK (I) < 31 THEN 19050: REM SPECIAL CHARS
19040 PRINT CHR$ ( PEEK (I));
19050 NEXT
19060 IF Z THEN PRINT
19070 START = START + SL + 2
19080 RETURN
20000 REM Error Handler
20010 SPEED= 255: TEXT : END

```